######################################################################
# Communi
######################################################################

TEMPLATE = app
TARGET = communi
DESTDIR = bin
DEPENDPATH += . src
INCLUDEPATH += . src
CONFIG += silent libircclient-qt
DEFINES += VERSION=\\\"0.2.0\\\"
QT += network script scripttools xml

OBJECTS_DIR = .tmp
MOC_DIR = .tmp
RCC_DIR = .tmp
UI_DIR = .tmp

QT_VERSION = $$[QT_VERSION]
QT_VERSION = $$split(QT_VERSION, ".")
QT_VER_MAJ = $$member(QT_VERSION, 0)
QT_VER_MIN = $$member(QT_VERSION, 1)

lessThan(QT_VER_MAJ, 4) | lessThan(QT_VER_MIN, 4) {
    error(Communi requires Qt 4.4 or newer but Qt $$[QT_VERSION] was detected.)
}

RESOURCES += communi.qrc
win32:RC_FILE = communi.rc
mac:LIBS += -l$$qtLibraryTarget(icuuc) -l$$qtLibraryTarget(icui18n) -l$$qtLibraryTarget(icudata)

include(src/src.pri)
include(translations/translations.pri)
